
HP-33S Calculator Program XYZ 1

Convert Latitude, Longitude and Height on any ellipsoid
to XYZ Geocentric Co-ordinates

Programmer: Dr. Bill Hazelton

Date: March, 2005.

Line Instruction Display User Instructions
X0001 LBL X Press  XEQ  X
X0002 6378137 Value of a for WGS84/NAD83/GRS80
X0003 STO  A
X0004 0.006694381 Value of e2 for WGS84/NAD83/GRS80
X0005 STO  E
X0006 INPUT  A 6378137 Enter value of a if different
X0007 INPUT  E 0.006694381 Enter value of e2 if different
X0008 INPUT  F Enter φ of point
X0009 INPUT  L Enter λ of point
X0010 INPUT  H Enter h of point
X0011 RCL  A
X0012 1
X0013 RCL  F
X0014 →HR
X0015 SIN
X0016 x2

X0017 RCLx  E
X0018 –
X0019 √x
X0020 ÷
X0021 STO  V
X0022 RCL+  H
X0023 RCL  F
X0024 →HR
X0025 COS
X0026 ×
X0027 RCL  L
X0028 →HR
X0029 COS
X0030 ×
X0031 STO  X
X0032 VIEW X X co-ordinate X co-ordinate is displayed
X0033 RCL  L
X0034 →HR
X0035 TAN
X0036 ×
X0037 STO  Y
X0038 VIEW  Y Y co-ordinate Y co-ordinate is displayed

X0039 RCL  V
X0040 1
X0041 RCL–  E
X0042 ×
X0043 RCL+  H
X0044 RCL  F
X0045 →HR
X0046 SIN
X0047 ×
X0048 STO  Z
X0049 VIEW  Z Z co-ordinate is displayed
X0050 RTN

Notes
(1) A program to convert latitude, longitude and ellipsoidal height on any ellipsoid to

X, Y, Z geocentric co-ordinates.

(2) The assumption is that the distances are in meters, but by using feet for the semi-
major axis of the ellipsoid, co-ordinates in feet will be produced.

(3) The program pre-enters the parameters for the WGS84/NAD83/GRS80 ellipsoid
by default (in meters), to save you having to remember these. If you want a
different ellipsoid, enter the appropriate a and e2 values at the prompts (A and E).

(4) The resulting co-ordinates are displayed with a prompt or label. Note that the
program does not clear registers after use. You can get ν for the point by using
the RCL V keystrokes, for example.

(5) The latitude and longitude are entered in HP notation, i.e., DDD.MMSS. The
height is assumed to be in the same units as the semi-major axis, by default,
meters.

(6) It is critical to follow the sign convention with latitudes, longitudes and heights.
Latitudes in the southern hemisphere are negative. Longitudes west of Greenwich
are negative, i.e., all longitudes in the US are negative. Heights below the
ellipsoid must be entered as negative.

Theory
The program implements the following four equations:

X  =  (ν  +  h)  cos φ  cos λ [1]

Y  =  (ν  +  h)  cos φ  sin λ [2]

Z  =  [ν (1 – e2) + h]  sin φ [3]

ν  =  

€ 

a
1 − e2 sin2 φ

[4]

These provide a direct solution. Values for the ellipsoid (a and e2) are requested, although default
values for NAD83/WGS84/GRS80 are supplied (these can be overwritten). The latitude,
longitude and ellipsoidal height of the point are requested of the user. The program supplies the
solution in three stages.

Sample Computation

Inputs a  =  6 378 137 m
e2  =  0.006 694 381 (WGS84/NAD83/GRS80 parameters)
φ  =  35°  00' 00" N  (entered as 35.0000) Latitude of point
λ  =  75° 00' 00" W  (entered as –75.0000) Longitude of point
h  =  200 m Ellipsoidal height of point

Results X = 1 353 776.483 m
Y = –5 052 362.616 m
Z = 3 637 981.622 m

Running the Program
Begin by pressing  XEQ  X

The calculator displays: A?
6,378,137.0000

This is the NAD83/WGS84/GRS80 ellipsoid semi-major axis. If this is OK, press R/S; if not key
in correct value and press R/S.

The calculator displays: E?
0.006694381    (suitably rounded, according to your settings)

This is the eccentricity of the NAD83/WGS84/GRS80 ellipsoid, e2. If this OK, press R/S; if not,
key in correct value and press R/S.

The calculator displays: F?
[Whatever value happens to be in this register]

Key in the latitude of the point and press R/S. Use negative values for the southern hemisphere.
In the given example, key in 35 and press R/S.

The calculator displays: L?
[Whatever value happens to be in this register]

Key in the longitude of the point and press R/S. Use negative values in the western hemisphere.
In the given example, key in –75 and press R/S.

The calculator displays: H?
[Whatever value happens to be in this register]

Key in the ellipsoidal height for the point and press R/S. Use negative values for heights below
the ellipsoid. In the given example, key in 200 and press R/S.

The calculator displays: X=
1,353,776.483 Press R/S

The calculator displays Y=
–5,052,362.616 Press R/S

The calculator displays: Z=
3,637,981.622

These calculations agree with the NGS website computations to within 0.003 m.

Storage Registers Used

A Semi-major axis of the ellipsoid, a. By default, set to  6378137 m

E Eccentricity of the ellipsoid, e2. By default, set to 0.006694381.

F Latitude (geodetic) of the point, φ.

H Ellipsoidal height of the point, h.

L Longitude of the point, λ.

V The radius of curvature of the ellipsoid in the prime vertical, ν.

X Geocentric X co-ordinate of the point.
Y Geocentric Y co-ordinate of the point.

Z Geocentric Z co-ordinate of the point.
Note that this program overwrites the Z register, removing the value of 360 that some other
programs use for correcting negative angles. Check this if using one of the closure programs.

Labels Used

Label X Length  =   198 Checksum  =  B137

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

Reference

The NGS website for the interactive XYZ ⇔ lat/long/height converter:

http://www.ngs.noaa.gov/TOOLS/XYZ/xyz.shtml


